
第 5章
表計算ソフトウエア：少し複雑な処理 (lookup関数，絶対参照，IF関数)

2023年 4月 19日

� �

� �

学習目標
(1) 絶対参照とスピルの考え方を理解する．
(2) ソートの考え方を理解する．
(3) IF関数を使って，条件分けをできるようになる．
(4) 表引き（LOOKUP関数）の考え方を理解する．
(5) さまざまな関数 (ROUNDUP,ABSなど）の使い方を理解する．
(6) 少し複雑な計算式になれる．
(7) 絶対参照，相対参照が混じったものやそのスピルでの計算方法を理解する．

本章は，専修大学商学部の高萩栄一郎の著作である．

1 表計算ソフトウエアのしくみ
計算式を複写するとき，複写元の計算式中セルの番地は，複写に伴う番地の移動に伴い，複写先の計算式中の番地は自動
的に変化します．絶対参照（または絶対番地）とは，複写に伴いセルが移動しても変化させない（固定する）番地の指定方
法です．逆に，セルの移動とともに変化させるセルの番地の指定を相対参照（または相対番地）といいます．プログラミン
グとの対応では，ループの中での定数が絶対参照，ループの中の配列の要素が相対参照に対応します．
表計算ソフトウエアは，複写すると相対参照でアドレスが調整されます．しかし，それでうまくいかないことがあります．
そのときは，上記の絶対参照という機能とその範囲の計算式をまとめて設定するスピルという機能があります．

2 複写のしくみ
図 1は，GDPの表を複写で設定した計算式で，次のことがわかります．

S2:T12(図 1の青の部分) 式がかかれているセルの数字が 1増えると，計算式中の数字も 1増えています．たとえば，S2

の計算式は「=SUM(B2:R2)」になっており，下に複写した S3の計算式は「=SUM(B3:R3)」と，すべて数字が 1増え
ています．

B12:R13(図 1の赤の部分 (一部)) 式が書かれているセルのアルファベットが 1つ増えると，右に複写された計算式のアル
ファベットも 1 つ増えています．たとえば，P13 の計算式は「=AVERAGE(P2:P11)」になっており，右に複写した
Q13の計算式は，「=AVERAGE(Q2:Q11)」とアルファベットが 1つ増えています．

したがって，「複写」という機能は，複写元のセルの番地と複写先のセルの番地との差の分，計算式中の番地を調整してい
ることがわかります．別の言い方をすれば，計算式は設定しているセルからの相対的な位置で記録され，これは相対参照と

複写するとアルファベットが増えていく

複

図 1 相対参照のしくみ (計算式を表示)

呼ばれています．たとえば，図 1の S2の計算式は，「=SUM(B2:R2)」になっています．この計算式は，

= SUM ( (自分より 17つ左のセル)から (自分より 1つ左のセル)まで)

というしくみになっています．多くの表は，この複写の番地調整機能（相対参照）でうまくいくのですが，うまくいかない
ことがあります．その例をやってみましょう．

■参考:計算式の表示方法 計算式を設定するとその計算結果（数値など）が表示されます．計算結果に替えて，計算式を表
示することもできます．



5–4

• メニューの 数式 →ワークシート分析の中の 数式の表示
• もう一度 数式の表示 をクリックすることにより元に戻る

3 絶対参照，相対参照とスピル
3.1 相対参照だけではうまくいかない例
図 2は，各支店の年度別の売上高を記入したものです．この表は，教科書のページからダウンロードできます (S6.xlsx)．
オレンジの背景色の部分は，計算式で入力する部分です

図 2 売上高の評価
伸び率を計算しましょう．伸び率の計算式は，

2011年の売上高− 2010年の売上高
2010年の売上高

で，表示形式は「パーセント」です．
次に，2011年の各支店の売り上げが，合計に占める割合（構成比率）を計算しましょう．

× B5: =B3/G3

http://www.isc.senshu-u.ac.jp/jtext/text2023/text_dl.html

となります．この計算式を B5から F5に，複写して見ましょう．おかしな結果になります．原因は表 1の「単純に複写した
ときの計算式」（相対参照）になってしまうからです．この場合，C5から F5の計算式がおかしいことがわかります．分子は
複写先のセルの番地に従って，セルの番地は，相対参照（番地調整機能）によってアルファベットが 1つずつ増えてよいの
ですが，分母の番地はいつも「G3」のままでなくてはなりません（青）．
これを解決するために，セル G3を変化させない方法 (絶対参照)と B5から F5の計算式をまとめて設定することにより，

G3を変化さえない方法があります（スピル機能）

表 1 セルの設定
セル B5 C5 D5 E5 F5

単純に複写したときの計算式 =B3/G3 =C3/H3 =D3/I3 =E3/J3 =F3/K3

正しい計算式 =B3/G3 =C3/G3 =D3/G3 =E3/G3 =F3/G3

3.2 絶対参照
表 1の正しい計算式のように，複写したときに G3を変化させないようにするのが絶対参照です．図 3のように，複写元

（B5）のセルを設定するとき，番地調整機能を働かせないようにします．これは，変化させないアルファベットや数字の前
に，「$」マークをつけます．したがって，複写元の計算式は，次式になります．

B5: =B3/$G3

この場合，G3は，常に G3で，数値の方は変化しないので，

B5: =B3/$G$3



5–6

としてもかまいません．$をつけた番地を「絶対参照」，$を付けない番地を「相対参照」といいます（図 3）．

図 3 絶対参照を使って設定した計算式

■絶対参照の指定の仕方 他の計算式と同様に，キーボードなどから$マークを入力することが可能ですが，セルをクリック
したあとに，簡単な操作で$マークを入力する方法があります．

• セルの番地 (G3など）をクリックなどで入力します．
• キーボードの� �

F4� �キーを押すことにより，$マークを付けることができます．
• $の位置が異なるときは，� �

F4� �キーを何回か押すことにより，指定の計算式にします．
動画:絶対参照（音声付き）・動画:絶対参照（音声なし）

http://www.isc.senshu-u.ac.jp/jtext/movies/S6_zettai.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies/S6_zettai__mute.mp4

3.3 スピルの利用
Excel 2021で追加されたスピルという機能で，絶対参照を使って指定した範囲の計算式をまとめて指定することができ
ます．まとめて式を設定する範囲を考えれば，絶対参照より直観的に理解できます．
スピルは，2023年度端末室に導入されているデスクトップ版の Excel 2019では利用できません．その場合，ブラウザー
版の Excelを利用してください．
スピルは，ある範囲のセルの計算をまとめて設定します．

■スピルの図解 図 4は，伸び率の計算を例です．図 4上段は，これまでのやり方でやったものです．

B4: =(B3-B2)/B2

複写元: B4

複写先: C4:G4

通常の利用ではこの方法を利用しましたが，スピルのイメージを把握するため，下段のようにスピルを使って設定します．
計算に利用する範囲は 2つ，計算結果 (出力範囲)が 1つになります．

範囲 P: B2:G2 2010年の値
範囲 Q: B3:G3 2011年の値
出力範囲 R: B4:G4 伸び率
スピルの計算式 (B4に設定) =(B3:G3-B2:G2)/B2:G2 B4に設定すれば，B4:G4に出力される

計算は， (2011年の値 - 2010年の値) / 2010年の値 ですので， (Q-P)/P となります．スピルでは，範囲の各値に対して順番
に計算していきます (各範囲の 1番目の値で計算した結果は，出力範囲の 1番目に出力)．出力範囲は，入力範囲に応じて自

=(B3-B2)/B2
コピー

計算式の複写

スピル

P
Q
R

( Qの範囲の各値 - Pの範囲の各値 ) / Pの範囲の各値
→ Rの範囲の各値

B4: =(B3:G3-B2:G2)/B2:G2

(Q-P)/P

図 4 スピルの説明

動的に拡張して表示されます（この拡張するのを，「こぼれ落ちる」ように増やすので，Spillと呼ばれています）．

■スピルの計算式の設定 (B4:G4) 計算式は，次のように設定します．

(1) B4 をクリックし,
� �
=� �,� �

(� �を入力します．
(2) マウスもしくはキーボードで B3:G3（範囲 Q）を範囲指定し,

� �
-� �を入力

(3) B2:G2（範囲 P）を範囲指定し,
� �
)� �� �

/� �を入力
(4) B2:G2（範囲 P）を範囲指定し,

� �
Enter� �を入力

(5) 計算結果が B4:G4（範囲 R）に表示されます．

動画:スピルによる設定 (伸び率)（音声付き）・動画:スピルによる設定 (伸び率)（音声なし）

■構成比率の計算 図 5は，スピルを使った構成比率の計算です．

TS

U
( Sの範囲の各値 ) / Tの値
→ Uの範囲の各値 B5: =B3:F3/G3

S / T

図 5 スピルによる構成比率の計算

範囲 S: B3:F3 2011年の値
セル T: G3 合計の値，範囲 Sの何番目かにかかわらず，いつも G3

出力範囲 U: B5:F5 構成比率
スピルの計算式 (B5に設定) =B3:F3/G3 B5に設定すれば，B5:F5に出力される

範囲 S(B3:F3, G3 は含めない) の値を　合計の値のセル (G3) で割って，S の各値の構成比率を求めます．構成比率は，

http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023A.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023A_mute.mp4

B3/G3 , C3/G3 ,. . .のように，範囲 S /セル T になります．分母は，範囲 Sの何番目かにかかわらずいつも G3です．スピ
ルの場合，１つのセルを指定した場合，いつもそのセルで計算します（これは，絶対参照で$を付けるのに対応します）．

B5: =B3:F3/G3

(1) まとめて計算式を指定する範囲の左上のセルを選択します (B5)

(2) 次の計算式を入力して（B3:F3は範囲指定を利用した方が便利），� �
Enter� �

B5: =B3:F3/G3

(3) B3:F5は，B3から右へ 5個のセルなので，計算式を入力した B5から F5までの 5個のセルにこぼれる (spill)よう
に,計算結果が表示されます．

動画:スピルによる設定 (構成比率)（音声付き）・動画:スピルによる設定 (構成比率)（音声なし）

■注意

•「#SPILL」「#スピル」というエラーが表示されることがあります．これは，拡張された出力範囲にすでに値が入力さ
れている場合に表示されます（その範囲を一度クリアするか，別の場所を指定します）．

• スピルで設定した計算式を変更するときは，左上のセルの計算式を変更します．

4 指数を計算
GDPの計算で，経済発展が大きい国はどこであるのか分析するとき，国の大きさに関わらず，GDPがどのように変化し
ているのかを分析することが重要です．そこで，物価指数のように基準年を設け，各国について基準年の値を 100としたと

http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023B.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023B_mute.mp4

き各年の GDPがいくつかになるのかを分析します．

図 6 指数化した表

図 6は，その計算表で新しく表（シート名：GDP index）を作成しました．計算式は，2001年のデータを 100として，そ
の何倍になったかを計算するので，

A国 t年の指数 (2001年基準) =
A国 t年の GDP

A国 2001年の GDP
× 100

となります．

■絶対参照による計算 B2の計算式は，ワークシート間の演算を利用して「=(GDP!B2/GDP!B2)*100」ですが，複写する
とき，2000年の GDPはいつも B列なので，Bを変化させないように Bの前に$をつけ，2の前には付けません．� �

F4� �キー

を使うとき分母の B2をクリックしたあと，� �
F4� �キーを 3回押すことにより!B2になります．2000年から 2007年までまとめ

て，計算式を複写します．

B2: =(GDP!B2/GDP!$B2)*100

複写元: B2

複写先: B2:O13

とします．B列も同じ計算式で，100となります．� �
F4� �キーは，1回押す毎に$マークが付く位置（アルファベットの前，数字

の前）を変更しますので，何回か押して正しい位置になるようにしてください．動画:Excel GDPの指数

■スピルによる計算 図 7は，スピルを使った各国地域の指数の計算です．
範囲 Vを範囲Wで割り 100倍する計算を指定しています．Vの各値をWのどの値で割るのかはこの指定では自動で調
整されます．具体的には，範囲 Xの各セルで，分子の範囲 Vのセルは，対応セルですが，分母の範囲Wでは 1列のみです
ので，範囲 Xの同じ行の計算式は共通に範囲Wの同じ行の値を使っています．これは絶対参照と同じで，範囲をまとめて
設定することにより$を付ける位置を自動で調整していると考えることもできます．

• ワークシート GDP indexを選択（または作成）
• まとめて指定する範囲の左上のセルを選択します (B2)

• B2に� �
=� �を入力

• GDPのワークシートで，B2:R12を範囲指定し，� �
/� �を入力

• GDPのワークシートで，B2:B12を範囲指定
•「*100」を入力し，� �

Enter� �
• 計算式は次のようになります．

http://www.isc.senshu-u.ac.jp/jtext/movies/Excel_GDPindex_a1.mp4

W

V

X
( Vの範囲の各値 ) / ( Wの範囲の各値 )*100 
→ Xの範囲の各値
※Vの同じ⾏のWは，対応するWの値を利⽤

GDP_index!B2: 
=(GDP!B2:V12/GDP!B2:B12)*100

(V / W ) *100

ワークシート：GDP_index

ワークシート：GDP 列E〜Sは省略

列E〜Sは省略

図 7 スピルを使った各国地域の指数の計算

B2: =GDP!B2:R12/GDP!B2:B12*100

• 計算式で「GDP!B2:R12」とあるので，B列から R列まで 17列，2行目から 12行目まで 11行，式を入力するセル
B2から右下へ 17列 12行まとめて計算します．

• 分母の「GDP!B2:B12」は行のみの変化ですので，行のみの範囲を指定します．
• B2:B12が，2001年基準（B列）を表します．

動画:スピル・指数の計算（音声付き）・動画:スピル・指数の計算（音声なし）

4.1 練習問題 S1-6

スピルまたは絶対参照を利用して作成しましょう．

(1) ワークシート「GDP index2008」を作成し，2008年基準（リーマンショック）の GDPの指数の表を作成しなさい．
(2) ワークシート「population index2001」を作成し，2001年基準の人口の指数の表を作成しなさい．
(3) ワークシート「GDPperCapita index2010」を作成し，2010年基準の一人あたり GDPの指数の表を作成しなさい．

5 練習問題
5.1 練習問題 S7-1

図 8は，関東の各都県の小学生，中学生，高校生，大学生の人数を千人単位で記入したものです．(出典：学校基本調査，
1999年 5月 1日現在)この表は，教科書のページからダウンロードできます (S7.xlsx)．

(1) B10から E10に，小学生，中学生，高校生，大学生の人数の合計を計算しなさい．
(2) F3に，茨城県の小学生人数の関東の都県の合計に占める割合（パーセント表示）を計算します

http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023C.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023C_mute.mp4
http://www.isc.senshu-u.ac.jp/jtext/text2023/text_dl.html

A B C D E F G H I

1 人数（単位：1000人) 各都県が占める割合

2 都県名 小学生 中学生 高校生 大学生 小学生 中学生 高校生 大学生

3 茨城 189 110 106 37 8.55% 8.94% 8.58% 3.20%

4 栃木 128 74 77 21 5.79% 6.01% 6.23% 1.82%

5 群馬 124 69 67 19 5.61% 5.61% 5.42% 1.65%

6 埼玉 408 217 206 114 18.45% 17.63% 16.67% 9.87%

7 千葉 339 190 183 114 15.33% 15.43% 14.81% 9.87%

8 東京 562 329 365 660 25.42% 26.73% 29.53% 57.14%

9 神奈川 461 242 232 190 20.85% 19.66% 18.77% 16.45%

10 合計 2211 1231 1236 1155 100.00% 100.00% 100.00% 100.00%

図 8 関東の県の児童生徒学生の比率

絶対参照またはスピルを使って指定しなさい．
絶対参照を使う場合 F3の計算式を F3から F9に複写するために，F3の計算式の適当な位置に$マークをつけます．
ただし，$マークは変化させないものの前につけます．
アルファベットを変化させたくない場合は，アルファベットの前，数字を変化させたくない場合は数字の前に$を
つけます．
F3の計算式を F3から F9に複写します．

スピルを使う場合 各都道府県の小中高大学生の合計を sum関数を使い，計算式の中に含めます．
スピルで，sum(範囲)を使うと，sum(範囲)は 1つの値を出力する関数ですので，1つの値として扱われます．

(3) 中学生，高校生，大学生の列も同様に計算式を設定します．

A B C D E

1 合否判定

2 合格基準点 70

3

4

5 番号 名前 得点 平均との差 合否

6 1 田中 100 34.8 1

7 2 大西 70 4.8 1

8 3 大久保 30 -35.2 0

9 4 前田 80 14.8 1

10 5 友田 65 -0.2 0

11 6 保田 77 11.8 1

12 7 大川 78 12.8 1

13 8 木村 32 -33.2 0

14 9 村田 54 -11.2 0

15 10 川上 66 0.8 0

16 合計 652 5

17 平均 65.2

図 9 合否判定

5.2 練習問題 S8-1

図 9のように入試の合否判定をします．この表は，教科書のページからダウンロードできます (S8.xlsx)．ただし，E列の
合否判定は，IF関数の節で学習します．

http://www.isc.senshu-u.ac.jp/jtext/text2023/text_dl.html

(1) 得点の合計，平均を計算しましょう．
(2) 平均との差を計算しましょう．平均値はいつも C17であることを注意しましょう．
絶対参照またはスピルを使います．

5.3 練習問題 S1-7

各国・地域の GDPが各年，10ヶ国・地域の合計に対しての割合（構成割合）を求めましょう．

(1) 新しいシート「GDP share」を作成し，指数と同様に，表頭，表側を複写します．
(2) A国・地域の t年の構成割合は，次式で計算します．

A国・地域 t年の構成割合 =
A国・地域 t年の GDP

t年の 10ヶ国・地域の GDPの合計
B2∼V11に，A国・地域 t年の構成割合を計算します．絶対参照またはスピルを使います．
絶対参照 B2 に計算式を設定します．B2 の場合，分子は 2001 年の Australia の GDP(GDP!B2) になります．分母
は，2001年の GDPの合計 (GDP!B12)になりますが，複写すとき，合計はいつも 12行目であることに注意しま
しょう．

複写元: B2 複写先: B2:R11

スピル 分子は，指数の計算では 2001年の値の列でしたが，構成割合では合計の行になることに注意しましょう．
(3) 表示形式をパーセンテージにします．
(4) 図 10は，計算例です．



5–18

図 10 GDPの構成割合 (一部)

6 ソート（並べ替え)関数
ソート（ソーティング，並べ替え）は，ある基準で行（または列）の順番を入れ換えることです．成績処理をするとき，得
点の大きい順に並べ替えるときなどに使います．GDPの例ではある年の GDPの大きさなどの順番で並べ替えてみます．例
題の表は，教科書のページからダウンロードできます (sort ex.xlsx)．

2023年度から本テキストでは Sort関数を利用します．従来のテキストでは，関数を使わずに，直接，表を書き換える方
法でした．直接書き換えると，分析の途中で元の表が書き換わり操作が複雑になったり，操作ミスにより元にデータが壊れ
たりしますので，ソート関数を使い．別の表に出力するようにしました．ソート関数はMS365でで利用できます．2023年
度の端末室 PCなど古い Officeの場合，ブラウザー版の Excelを利用してください．

http://www.isc.senshu-u.ac.jp/jtext/text2023/text_dl.html

6.1 ソート関数の利用
Excelではソートをする関数は，SORT関数と SORTBY関数の 2つあります．SORT関数は 1基準のみで，SORTBY関
数は複数基準を扱えます．本テキストでは，複数基準を扱える SORTBY関数を学修します．
ソート (並べ替え)は，ある基準に従って行を並べ替えます．並べ替えの方法は，大きい順は降順，小さい順 (または ABC

順，あいうえお順)は昇順と呼びます．並べ替えの基準の列ををソートキーもしくは単にキーといいます．
この節の例題は，教科書のページからダウンロードできます (sort ex.xlsx)．図 11の元データ（出席簿のデータ）で，クラ
ス，出席番号はばらばらの順番です．まず，真ん中の 1基準のように，クラスの順番（昇順）で並べ替えてみます．この場
合，クラスの順番は整いますが，出席番号はばらばらです．そこで，出席番号の大きい順（降順）に並べます（通常は，小さ
い順ですが，ここでは練習のため大きい順にします）．このように，基準となる列が複数指定するとき，基本はクラス順で，
クラスが同じ時とは出席番号順に並べ替えるとき，クラス順を１番目のソートキーにし，出席番号を２番目のソートキーに
します．

SORTBY関数は，元のデータの範囲を指定し，別の範囲にソートした結果を出力します．例えば，図 11のワークシート
元データ の範囲 A2:A9を別の範囲，図 11のワークシート 1基準 の A2:C19に，ソート結果を表示します．関数は，左上
のセル A2に指定すれば，C19まで自動で拡張 (スピル)して表示されます． 1基準 の A2:C19のように，スピルの範囲は
スピル範囲と呼ばれています．

SORTBY関数は次のように指定します．

= SORTBY（もとの範囲,基準 1の列の範囲,基準 1の順序,基準 2の列の範囲,基準 2の順序, . . .）

元の範囲 並べ替える元の範囲（図 11の例では， 元データ の A2:C19）．
基準 1の列の範囲 並べ替える基準の列の範囲．行数は並べ替える元の範囲と同じ行数（図 11 の 1基準 の例では，

http://www.isc.senshu-u.ac.jp/jtext/text2023/text_dl.html

元のデータ 1基準（クラス昇順) 2基準(クラス昇順,出席番号降順)

図 11 ソート関数の図解

元データ の A2:C9）．
基準 1の順序 昇順の場合 1，降順の場合-1を指定

図 11のワークシート 1基準 の A2の計算式を設定します．

(1) ワークシート 1基準 の A2をクリックします．
(2) =「=SORTBY(」と入力
(3) = 元データ の A2:C19を範囲指定し，「,」を入力

(4) = 元データ の A2:A19を範囲指定し，「,」を入力
(5) =アルファベット順は昇順なので，「1」を入力
(6) =「)」を入力し，� �

Enter� �を入力
=SORTBY(元データ!A2:C9,元データ!A2:A9,1)

となります．
動画:Sortby関数，１基準（音声付き）・動画:Sortby関数，１基準（音声なし）
ワークシート 2基準 の A2 の計算式は，次のようになります．基準１と同じ方法で設定してみましょう．手順の (3),(4)

をくり返すのと，基準 2の順序が降順であることに注意してください．

=SORTBY(元データ!A2:C9,元データ!A2:A9,1,元データ!B2:B9,-1)

ソート関数の特徴

• ソート関数は，元の表を書き直すと自動的にソート関数で出力されるスピル範囲は変化します．例えば， 元データ
のクラス記号や出席番号を変更してみましょう． 1基準 や 2基準 の表が変化します．

• スピル範囲の個別のセルは変更できません．
計算式を変更するには，左上のセル (A2)の計算式を変更します．
スピル範囲を消すには，スピル範囲全体 (A2:C9) の中をクリックすると青色の枠で選択されるので，そこで� �

Delete� �
キーで削除します．

動画:Sortby関数，2基準（音声付き）・動画:Sortby関数，2基準（音声なし）

http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023D.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023D_mute.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023E.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023E_mute.mp4

6.2 練習問題 S1-8

ワークシート「GDP Sort」を作成し，2001年を基準として大きい順にソートした結果を表示する表を作成しなさい．
※ソートする範囲は，列はデータの一番右まで含め，行は見出し行や合計や平均の行は含めません（含めると一緒にソート
します）．

7 RANK関数を利用して順位を求める
S1.xlsxのワークシート GDP の AC列のように，2001年の GDPでの順位を求めましょう（図 12）．順位は，RANK関数
という関数とスピルという Excelの新しい機能で求めることができます．
大学の PCの端末室に設置されている PCなどで利用するときは，ブラウザー版の Excelを利用しましょう．
※　表の画面をを書き直す必要あり

7.1 RANK関数
図 13は，RANK関数の図解です．RANK関数には，3つの引数が有ります．

数値 順位を求める値です．
範囲 この範囲で指定した範囲の順位を求めます．
順序 　大きい順（降順）のとき 0，それ以外（小さい順 (昇順））のとき 0以外の値を指定します．

図 12 RANK関数（ワークシート GDP）

=RANK(B2,B2:B11,0)

数値:次の範囲内で
の順位を求める値

範囲：この範囲の順位を求める

下のセルを含めて，この範囲

順序
大きい方が高順位→ 0
小さいとき高順位→ 0以外の値

図 13 RANK関数の図解

7.2 スピルを使って設定
計算式は，次のようになります．

Y2: =RANK(B2:B11,B2:B12,0)

• RANK関数の最初の引数は，１つの値なので，最初の引数の B2:B11は，下に 10個の分 (Y2:Y11)に計算することを
指定します．

• RANK関数の 2つ目の引数は，順位を求める範囲なので B2:B11とします．(特に，「$」を付ける必要はありません)

図 13は RANK関数の図解で RANK関数は 3つの引数を使います．AC2にオーストラリアの 2001年の順位を求めます．
次のように設定します．

(1) AC2をクリック
(2) 「=RANK(」と入力
(3) B2から B11を範囲指定し「,」を入力
(4) B2から B11を範囲指定
(6) 「）」を入力し，� �

Enter� �を入力
(7) AC2から AC11に 2001年の順位が表示されたと思います．

動画:RANK関数スピル（音声付き）・動画:RANK関数スピル（音声なし）

http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023F.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023F_mute.mp4

7.3 絶対参照を使って設定
次のように設定します．

(1) AC2をクリック
(2) 「=RANK(」と入力
(3) B2　をクリックし，「,」を入力
(4) B2から B11をマウスで範囲指定
(5) キーボードから� �

F4� �キーを 1回入力．「B2:B11」が「$B$2:$B$11」となります．
(6) 「）」を入力し，� �

Enter� �を入力
複写　

複写元: AC2 　　 複写先: AC3:AC11

動画:Excel順位を求める,関数の挿入を利用

8 練習問題
8.1 練習問題 S1-9

(1) AD列に 2021の順位を求めましょう．

http://www.isc.senshu-u.ac.jp/jtext/movies/Excel_rank_a1.mp4

(2) ワークシート GDPperCapitaの AC列に 2001年の，AD列に 2021年の一人あたりの GDPの順位を求めましょう．

8.2 練習問題 S4-2

練習問題 S4-1で作成したモーグルの表を「トータルスコア」の高い順に順位を付けなさい．

9 オートフィルター
オートフィルタは，ある特定の条件を満たす行だけ抽出する機能です．

9.1 オートフィルターの設定・解除
オートフィルターを実行すると，大きい順や小さい順に並べ替える機能がありますが，表を書き換えてしまうので利用し
ないようにしてください．（ソートした結果は，フィルーターを解除しても戻らないようですので，設定した場合� �

Ctrl� �+� �
z� �で

元に戻してください．

(1) 項目名の列を範囲してします．GDPの例では，（例：A1:X1)を範囲指定します．
(2) メニューの データ → 並べ替えとフィルター の中の フィルター
(3) A1:X1の右に， ▼ のボタンが表示されます．

■フィルターを使って，行の指定 次に，フィルターで扱う行を制限します（合計や平均の行を扱わないようにします）

(1) 項目名の右の ▼ のボタンをクリックします（ここでは，A1の右の ▼ ボタンをクリック）

(2) メニューが表示されるので，扱わない行（ここでは，「合計」，「平均」，「空白のセル」のチェックを外します．
(3) OK をクリック

■抽出 オートフィルターを使うことにより，必要な行だけ抽出することができます．たとえば，2017年の GDPが 10000

百万ドルより多い行だけ選択してみます．

(1) 基準の列（2017の列）の ▼ のボタンをクリックします
(2) 数値フィルターをクリックします，
(3)「指定の値より大きい」を選び，10000を入力し， 完了 をクリック

■フィルターの解除 使い終わったら，フィルターの解除を行ってください．

(1) メニューの データ → 並べ替えとフィルター の中の クリア

※フィルターの機能に，並べかえの機能がありますが，これを利用すると表を書き換えてしまうので，この演習では利用し
ないようしましょう．

10 IF関数
10.1 IF関数とは

S6の売上高評価の表で，「伸び率評価１」の計算式を設定しましょう．これは，伸び率が正だったら 1，0または負だった
ら，0を表示させます．
あるときは，こうして，そうでないときは，こうするといったことをしたいときは，IF関数を使います．使い方は，

図 14 売上高の評価 (再掲)

IF(条件,条件を満たしているときすること,条件を満たしていないときすること)

となります．したがって，

B6: =IF(B4>0,1,0)

とします．その意味は，図 15のようになります．
大小関係などの条件の表記は，数学の表記とは若干異なり，表 2のようになります．
練習として，伸び率評価２の計算式を設定しましょう．伸び率評価２は，その支店の伸び率が全体の伸び率（セル G4の
値）より大きいまたは等しければ 1を，そうでなければ 0を表示させます．絶対参照またはスピルを使いますので注意して
ください．

10.2 講演会参加費
図 16は，ある講演会の会費の計算をしたものです．この表は，教科書のページからダウンロードできます (S9.xlsx)．

http://www.isc.senshu-u.ac.jp/jtext/text2023/text_dl.html

図 15 IF関数の図解

表 2 Excelでの大小関係の表記
大小関係 Excelの表記

= =

̸= <>

> >

≥ >=

< <

≤ <=

会員／非会員 (B列) 会員と非会員では，参加費が異なります．会員には，「M」と入力されています．
学生／非学生 (C列) 学生には，参加費が割引（学生割引）されます．学生には，「S」と入力されています．
パーティ (D列) 講演会終了後に開かれるパーティに参加する人は，「P」と入力されています．

A B C D E F G H

1

ＮＯ 会員／非会員 学生／非学生 パーティ 参加費 学生割引 パーティ代 合計

2

1 M S P 6,000 3,000 4,000 7,000

3

2 S 9,000 3,000 0 6,000

4

3 P 9,000 0 4,000 13,000

5

4 M P 6,000 0 4,000 10,000

6

5 M S 6,000 3,000 0 3,000

7

6 M S 6,000 3,000 0 3,000

8

7 M S 6,000 3,000 0 3,000

9

8 M P 6,000 0 4,000 10,000

10

9 9,000 0 0 9,000

11

10 M 6,000 0 0 6,000

12

11 M P 6,000 0 4,000 10,000

13

12 M 6,000 0 0 6,000

14

13 M 6,000 0 0 6,000

15

14 M S P 6,000 3,000 4,000 7,000

16

15 M S 6,000 3,000 0 3,000

17

16 M P 6,000 0 4,000 10,000

18

17 P 9,000 0 4,000 13,000

19

18 9,000 0 0 9,000

20

19 M P 6,000 0 4,000 10,000

21

20 M P 6,000 0 4,000 10,000

22

合計 135,000 21,000 40,000 154,000

23

24

参加費（会員） 6,000

25

参加費（非会員） 9,000

26

学生割引 3,000

27

パーティ 4,000

図 16 講演会会費

参加費 (E列) 参加費を記入します．参加費は変更されることもあるので，C24 　もしくは　 C25 　を参照するようにし
ます．

学生割引 (F列) 学生割引を記入します．学生割引は変更されることもあるので，C26を参照するようにします．

パーティ代 (G列) パーティ代を記入します．パーティ代は変更されることもあるので，C27を参照するようにします．
合計 参加費−学生割引+パーティ代　で計算します．

■参加費の計算（スピル）

E2: =IF(B2:B21=”M”,C24,C25)

動画:スピル・参加費の計算（音声付き）・動画:スピル・参加費の計算（音声なし）

■参加費の計算 (絶対参照) E2のセルの計算式を設定します．参加費は，B2が，「M」か空白によって異なるので IF関数
を使います．

△ E2: = IF(B2=”M”,6000,9000)

B2 が「M」かどうか調べるのに，「B2=”M”」とします．文字は，””で囲みます．M のとき，6000 なので，真の時の値を
6000とし，偽のときの値を 9000としました．しかし，その 6000と 9000という値は，変更されることがあり，C24と C25

を参照するようにします．C24,C25は，下に複写されたときも固定される番地なので，絶対参照にします．したがって，

E2: =IF(B2=”M”,$C$24,$C$25)

とし,下方向に複写します．

10.3 練習問題 S9-1

S9(参加費の計算)を完成させましょう．

http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023G.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023G_mute.mp4

10.4 練習問題 S8-2

S8の合否を設定します．合格表示という列に，基準点以上であれば 1を，そうでなければ 0を表示するようにします．も
ちろん，合否の列は 1箇所だけ計算式を作成して，あとは複写で計算式を設定してください．合格表示の合計を算出すれば
合格者数を求めることができます．
基準点を変えて，何人合格になるのか試してみようと思います．つまり，基準点を変えてみて，合格者が何人になるのか
シミュレーションを行います．

11 表引き (LOOKUP関数)

コンピュータを利用しているとよくコードを使います．例えば，大学生なら学生コード，商品には商品コードが付けられ
ています．また，ある範囲での対応表から表を引くという作業があります．例えば，所得額から，税額を求めるといったこ
とです．これらの機能を実現するのが VLOOKUP(または，HLOOKUP)という関数です．

JR南武線 (路線図)の運賃計算の例で説明しましょう．図 17は，南武線の２つの駅名を入力して，川崎からの距離を求め，
その差から，2駅間の距離を求め，それから運賃を求めるというものです．この表は，テキストのページからダウンロード
できます (S101.xlsx(運賃計算)，南武線は，電車特定区間の運賃）．
図 17の表は 3つの表からできています．一番上が運賃を計算する表，二番目が駅名と川崎からの距離の対応表，三番目
は，距離と運賃の対応表です．手順は次のようになります．

(1) 乗車駅名から川崎からの距離を求めます（C2)．
(2) 降車駅名から川崎からの距離を求めます (C3)．
(3) 2つの「川崎からの距離」の差をもとめ，小数点未満を切り上げます (C4,C5)．

http://ja.wikipedia.org/wiki/%E3%83%95%E3%82%A1%E3%82%A4%E3%83%AB:%E5%8D%97%E6%AD%A6%E7%B7%9A%E8%B7%AF%E7%B7%9A%E5%9B%B3.png
http://www.isc.senshu-u.ac.jp/jtext/text2021/text2_dl.html

図 17 JR南武線の運賃計算

(4) 距離と運賃の対応表から運賃をもとめます (C6,C7)．

11.1 乗車距離から運賃を求める
まず乗車距離から運賃 (C6)を求めてみましょう．仮に乗車距離として C5に 4を入力します．C6を求めるには，距離と
運賃の対応表を C5の 4について引くことです．乗車距離は 4なので表のロの行「4　 154」の行があてはまり運賃は 154

円になります．

図 18 JR運賃表と Excelの VLOOKUP関数用の表との対応関係

11.1.1 対応表の見方
図 18は，JR運賃表を Excelの VLOOKUP用の表には，どのように記述したらよいのかを示しています．Excelの表の各
行がどの数値をあてはまるかは，その行の距離の値から，次の行の距離の値までです．正確に言うと対応表の各行にあては
まる数値は，その行の距離から次の行の値未満までです．

距離が 1の行（イ） 距離が 1以上 4未満の運賃が 133円であることを示しています．
距離が 4の行（ロ） 距離が 4以上 7未満の運賃が 154円であることを示しています

例えば，乗車距離が 22であったら，「21　　　 388」の行（ヘ）にあてはまります．

11.2 VLOOKUP関数
C6の計算式は，図 19は，VLOOKUP関数の図解です．

C6: =VLOOKUP(C5,A21:C30,2,TRUE)

VLOOKUP関数は，VLOOKUP(照合値,照合範囲,列位置,照合方法)の 4つの引数を書きます．(C5に何も書いていないと
エラーになるので，C5に仮の 2駅間距離 (例えば 6)を入力し，この値をいろいろ変えて見てください．）

照合値 表引きをする表で照合する値．この例では，乗車距離 (C5)

（距離と運賃の対応表の一番左の列から探す値）
照合範囲 (対応表の範囲) 照合する表，ただし，照合値の列は，照合する範囲の一番左の列になるようにします．
列位置 照合した行の左から何番目の列を表示するか．一番左の列を 1列目と数えます．
照合方法 完全に一致するもののみを探すのか (FALSE)，下の行と未満のものも探すか (TRUE)を指定します．

=VLOOKUP(C5,A21:C30,2,TRUE)

照合値：
探す値

照合範囲：A21:C30
探す範囲（対応表の範囲）

1列目 2列目 3列目

列位置：
対応する行の何列目を
表示するか

照合方法：
対応表の行と行の間を探すか

CBA
川崎からの距離駅名1

20.8稲田堤乗車駅2
17.3登戸降車駅3
3.5２駅間距離 切り上げ前4
4２駅間距離 切り上げ後5

154運賃(ICカードを利用)6
運賃(切符を利用)7

CBA
運賃(切符)運賃(IC)距離20

140133121
160154422
170165723
2202161124
3103021625
3903882126
4704642627
5505503128
6406373629

4130

図 19 VLOOKUP関数の書き方

照合方法が TRUEのとき，表引きをする表は，一番左の列をソートキーに昇順 (小さい順)にソートされてなくてはなりませ
ん．理由は，照合するとき，照合値が，一番左の列の値と次の行の一番左の列の値の間にあるかどうかで，照合するかしな
いか決めているからです．

C6: =VLOOKUP(B5,A21:C30,2,TRUE)

練習として，C7の切符を利用した場合の運賃を計算しましょう．この場合 3列目になることに注意しましょう．C5の仮
の値をいろいろ変更してみて，それぞれの運賃が正しく表示されているか確認しましょう．
動画:スピル・Vlookup関数（音声付き）・動画:スピル・Vlookup関数（音声なし）

11.3 駅名から距離を求める
次に，駅名から「川崎からの距離」を求める関数を求めましょう．駅名から距離を求めるときは，完全に一致する行のみを
探します．したがって，ソートしておく必要はありません．C2の計算式は，B2を照合値として，A10:B18の照合範囲を探
し，1列目と等しければ，2列目の値を返すというものです．完全に一致する行のみ探すので，「FALSE」にします．C3も
同様の計算式のため，絶対参照もしくはスピルを利用します．

C2と C3は，基本的に同じ計算式で，照合値が B2か B3かの違いで，照合範囲は同じで，列位置，照合方法は一定の値
です．

■スピルを利用するとき

C2: =VLOOKUP(B2:B3,A10:B18,2,FALSE)

範囲 B2:B3（縦１列，２個）の値を照合値をスピルの範囲として使い，照合範囲 A10:B18は共通で使います．出力は，照合
範囲が縦 1列，2個（2行）ですので，出力のスピル範囲も C2:C3の縦 1列，2個（2行）になります．
動画:スピル・駅名から距離（音声付き）・動画:スピル・駅名から距離（音声なし）

■絶対参照を利用するとき 絶対参照を利用するときは，照合範囲が常に A10:B18 になるので, $A$10:$B$18 のように，
絶対参照にし，C2を C3に複写します．

http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023H.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023H_mute.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023I.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023I_mute.mp4

C2: =VLOOKUP(B2,$A$10:$B$18,2,FALSE)

複写元: C2

複写先: C3

11.4 2駅間距離を求める（絶対値，切り上げ）
2つの駅間の距離を求めるには，2つの駅の川崎からの距離の差を求めます．負の値になることがあるので，絶対値をとり
ます．絶対値をとる関数は，ABSです．まず，2駅間距離切り上げ前は，

C4: =ABS( C2 - C3 )

となります．JRの運賃計算は，小数点以下の端数は切り上げです．そこで，C4の値を切り上げて，C5に表示します．

C5: =ROUNDUP(C4,0)

となります．ROUNDUP(A,B)で，Aは切り上げる数，Bは切り上げる単位を示しています．0で小数点の位置，1で 0.1単
位，2で 0.01単位，-1で 10の単位で切り上げです．同様に，ROUNDが四捨五入，ROUNDDOWNが切り捨てです (表 3)．

表 3 切り上げ，四捨五入，切り捨て

ROUNDUP(3.14,0) → 4 ROUNDUP(586,-1) → 590 ROUNDUP(16.58,1) → 16.6

ROUND(3.14,0) → 3 ROUND(586,-1) → 590 ROUND(16.58,1) → 16.6

ROUNDDOWN(3.14,0) → 3 ROUNDDOWN(586,-1) → 580 ROUNDDOWN(16.58,1) → 16.5

12 練習問題
12.1 練習問題 S102-1

図 20 は，小田急ロマンスカーの新宿からの運賃，特急料金の表です (路線図)．B14 に駅名を入力したら，B15 の運賃，
B16の特急料金を計算する式を設定しましょう．この表は，テキストのページからダウンロードできます (S102.xlsx(ロマン
スカー)）．

� � �

図 20 ロマンスカーの運賃・料金

ヒント：B16を計算するとき，A3:C11までを照合範囲として，Vlookup関数を使います．列位置を設定することで特急
料金を表示させます．

http://www.odakyu.jp/romancecar/station/
http://www.isc.senshu-u.ac.jp/jtext/text2021/text2_dl.html


5–40

12.2 練習問題 S4-4

S4.xlsxのモーグルで，「エア点」の計算と「スピード」の計算では，小数第 3桁以下を切り捨て，小数点以下第 2位まで
求めるように変更ししましょう．

12.3 練習問題 S103-1

図 22は，表 21にしたがって．得点から，S,A+,A,B+,B,C+,C,Fの評価を計算する表です．VLOOKUPで使う得点と評価
の対応表（照合範囲）を作成し，C2～C11の計算式を設定しなさい．この表は，テキストのページからダウンロードできま
す (S103.xlsx(成績評価)）．

得点 評価
90～100 S

85～89 A+

80～84 A

75～79 B+

70～74 B

65～69 C+

60～64 C

0～59 F

図 21 大学の成績評価

図 22 得点から成績を求める

http://www.isc.senshu-u.ac.jp/jtext/text2021/text2_dl.html

• VLOOKUP関数で照合方法が TRUE(行と行との間も検索する場合)のとき，照合値の列（一番左の列）が昇順（小さ
い順）でソート（並べ替え）されていなくてはなりません．したがって，表 21に対応する VLOOKUPの照合範囲の
表は，表 21の行の順番とは逆になります．「0から 59」の行が 1行目，「90～」の行が最下行になります．

13 絶対参照と相対参照の混合とそのスピルでの計算
図 23 は，ネットショッピングの支払いを各通貨別に集計したものです．換算レートは，2012 年 3 月現在のものです．
赤の部分は計算式で求める部分です．この表は，テキストのページからダウンロードできます (S104.xlsx(ネットショッピ
ング)）．
表には，日付と品名，支払った通貨の種類（GBPはイギリスポンド，EURはユーロ，CHFはスイスフラン，USDはアメ
リカドルをあらわします．），その通貨での金額を記入します．表の右側で，通貨毎に集計し，換算レートをかけて，日本円
（邦貨）の金額を求めます．日付の入力では「8月 10日」は「8/10」と入力します．自動的に，「8月 10日」に変換されます．
日付としてうまく表示されない場合や，日付から数値の表示形式に変更するときは，右クリックして， セルの書式設定
の 表示形式 のタブで変更します．
最終的には，図 23のような表にして，各通貨毎の合計等を求めます．
まず，計算式を設定しましょう．まず，E4について考えます．

E4は，もし，この行の通貨が，GBPであったら，この行の金額の欄の値を表示する．
↓

E4は，もし，C4と E3が等しかったら，D4を表示し，そうでなかったら，空白を表示する
（空白は，ダブルコーテーション「”」を 2つ重ねます）．

↓

http://www.isc.senshu-u.ac.jp/jtext/text2021/text2_dl.html

図 23 ネットショッピング支払い

E4: =IF(C4=E3,D4,””)

となります．次に，この計算式を複写してみましょう．

複写元: 　 　 E4 　　 複写先: 　 　 E4:H12

としてみましょう．間違っていることがわかります．
このような場合，2つの方法で解決できます．1つはスピルを使った方法で，もう 1つは絶対参照と相対参照の混合させ
る方法です．

13.1 スピルを使った計算
スピルを使っての計算式の場合，E4の =IF(C4=E3,D4,””) の IF関数の各部分について範囲 E4:H12の全セルに対して，
図 24のように考えます．範囲 D(出力の範囲)は，範囲 Aと範囲 Bの値が等しいとき，範囲 Cの値を出力します．そうでは
ないとき，””を出力します．

A
B

C D A = B だったら
C

そうではなかったら
"" (空⽩）

E4:   =IF(C4:C12=E3:H3,D4:D12,"")

Dの計算式

図 24 スピルによる計算，ネットショッピング支払い

IFの条件部 この部分は，C 列 (C4:C12 範囲 A) と　 3 行目 (E3:H3 範囲 B) の値が等しかったらという条件になるので，
C4:C12=E3:H3 とします．

条件を満たしているとき この部分は，D列 (D4:D12範囲 C)の値を表示するので， D4:D12 とします．
条件を満たしていいないとき (””) そのまま ”” とします

E4: =IF(C4:C12=E3:H3,D4:D12,””)

とします．
動画:スピル・ネットショッピング（音声付き）・動画:スピル・ネットショッピング（音声なし）

13.2 絶対参照の利用 (絶対参照と相対参照の混合を使った計算)

E5,E6の計算式（表 4上部）を見てみましょう．E5の計算式は，IF(C5=E4,D5,””)になっています．しかし，正しい計算
式は，IF(C5=E3,D5,””)です．E6,E7も表のようになります．原因は，E4の式中の E3は，下に複写するとき，数字の部分
が増えてはいけないのに，番地調整機能によって，増えてしまっているからです．

表 4 絶対参照と相対参照の混合
セル 単純に複写したときの計算式 正しい計算式
E4: =IF(C4=E3,D4,””) =IF(C4=E3,D4,””)

E5: =IF(C5=E4,D5,””) =IF(C5=E3,D5,””)

E6: =IF(C6=E5,D6,””) =IF(C6=E3,D6,””)

E7: =IF(C7=E6,D7,””) =IF(C7=E3,D7,””)

E4: =IF(C4=E3,D4,””) =IF(C4=E3,D4,””)

F4: =IF(D4=F3,E4,””) =IF(C4=F3,D4,””)

G4: =IF(E4=G3,F4,””) =IF(C4=G3,D4,””)

H4: =IF(F4=H3,G4,””) =IF(C4=H3,D4,””)

E4の計算式を右に複写したものを表 4下部に示します．赤の部分が間違っています．これは，C4の部分と D4の部分の

http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023J.mp4
http://www.isc.senshu-u.ac.jp/jtext/movies2/movieschap5/Spill2023J_mute.mp4

アルファベットは，横に複写されるときに，増えてはいけないのに番地調整機能により，増えてしまったからです．
複写のとき，番地調整機能を働かせないように「$」を付けます．上の例では，

　　　 E4: =IF($C4=E$3,$D4,””)

となります．複写しても，Cと 3と Dは，変化せずいつも Cと 3と Dです．E4を設定したら，他の部分に複写します．

複写元： E4 　　 複写先： E4:H12

13.3 日本円への換算等
合計などを計算しましょう．換算レートは，1 海外通貨あたり日本円でいくらかを示しています．例えば，155 ポン
ド (GBP) 使い，換算レートが 1 ポンド=132.154 円のときの邦貨金額 (日本円金額) は，155 ポンド × 132.15 円/ポン
ド = 20, 483円となります．I15は，邦貨金額の合計です．15行目はセルの書式設定で，小数点以下を四捨五入しましょう．
通貨記号の欄を変更したり，現在の為替レート（ 朝日新聞デジタル：為替 -経済・マネー )に直したりしても正しく動作
するか確認しましょう．

14 練習問題
14.1 1次関数 3本のグラフ化

1次関数 f(x) = ax+ bで，aと bを変えた 3本をグラフ化する表を作成しましょう．図 25左の B2:D3に 3本の 1次関
数の aと bの値を記入しておきます．ただし，計算式を設定する部分は，次のようにします．

(1) A6に −10を入力し，A7に「=A6+0.5」とします．

http://www.asahi.com/business/exchange/


5–46

(2) B6に計算式を設定し，その式を B6:D46に複写します．B6の計算式は，絶対参照と相対参照が混合していますので
注意してください．

図 25 y = ax+ bのグラフ化のための表とグラフ

(3) 図 25右のような 3本のグラフ (散布図（直線）)を作成しましょう．
(4) それぞれの直線のパラメータである a, bを変化させたとき，どのように変化するのかを確かめましょう．

14.2 練習問題 S105-1
図 26 は，ある架空の電話会社の電話料金の計算表です．この表は，テキストのページからダウンロードできます

(S105.xlsx(電話料金)）．

• 10円で話せる時間は，通話区分，距離で決まります．
• 通話区分は 3区分：「平日の昼間」(区分 1)，「平日の夜または休日の昼夜」(区分 2)，「平日，休日の深夜」(区分 3)

http://www.isc.senshu-u.ac.jp/jtext/text2021/text2_dl.html

図 26 電話料金の計算

• 距離は，通話先までの距離で，表の 14行目の 0は，0km(隣接）以上 20km未満を表しています．また，19行目の距
離 170の行は，170km以上すべてを表しています．

• たとえば，通話区分が 2で，70kmの場合，60の行の通話区分 2の列 (3列目）の場合の 10円で話せる秒数は 45秒
になります．

• 料金は，まず度数を，

度数 =
通話時間（秒）

10円で話せる秒数
で計算し (F列)，小数点以下を切り上げます (G列，切り上げの計算方法は，表 3を参照)．

• 1度数は 10円なので，度数の列 (G列)に 10をかけたものが電話料金になります．

では，図 26の上方のエリアを使って電話料金の計算表を作成しましょう．
ヒント

•「10円で話せる秒数」は，距離と通話区分を使って，料金表 (A14:D19)から VLOOKUPを使って探します．そのと
き，通話区分が 1のとき 2列目，2のとき 3列目 . . .であることから，VLOOKUPの列位置は，通話区分のセルの値
を使った計算式になります (通話区分の値に 1加えた数が列番号になっています)．

•「度数 (切り上げ前)」は，単純に， 通話時間（秒）
10 円で話せる秒数 を計算したものです．

•「度数 (切り上げ後)」は，「度数 (切り上げ前)」の値を切り上げたものです．

完成したら，通話区域，通話の種類や時間を変えてみて，正しく計算されているかどうかチェックしましょう．


